quarter wavelength long. The input power, fed into node (1), is split up with half propagating clockwise through the ring and the other half propagating counter-clockwise. The resulting voltages at nodes (2) and (3) occur with a 90° phase difference, provided that the ports are terminated with a resistor equal to characteristic impedance. Node (4) has zero voltage because the clockwise and counter-clockwise portions have a 90° and 270° phase shift, respectively, and will therefore cancel each other. In the graph (below) the frequency response of the voltages at nodes (1) through (4) are shown.




```
HYBDRING - 1/4 Wavelength Hybrid Coupler
R1 5 1 50
R2 2 0 50
R3 3 0 50
R4 0 4 50
X1 1 2 3 4 RING
V0 5 0 AC 2
.AC LIN 100 3.3GHZ 5.3GHZ
.PRINT AC VDB(1) VDB(2) VDB(3) VDB(4)
.PRINT AC  V(1)  VP(1)  V(2)  VP(2)
.PRINT AC  V(3)  VP(3)  V(4)  VP(4)
.SUBCKT RING 1 2 3 4
T1 1 0 2 0 Z0=35.4 F=4.3GHZ NL=.25
T2 2 0 3 0 Z0=50.0 F=4.3GHZ NL=.25
T3 3 0 4 0 Z0=35.4 F=4.3GHZ NL=.25
T4 4 0 1 0 Z0=50.0 F=4.3GHZ NL=.25
.ENDS
.END
```



IsSpice AC Analysis of HYBDRING

6.1.2 Hybrid Transformer

In the lower high frequency region, the hybrid coupler is ineffective because its branches become too long. However, the connection of subsystems without mutual interaction can be realized with conventional circuits that are built with discrete components. In the hybrid transformer, power is injected into node (1) and splits up equally to nodes (3) and (4). Node (2) has zero voltage and is completely isolated from node (1). The same is valid when interchanging the ports cyclically. The largest possible bandwidth is obtained for trifilar windings, distributed uniformly around the periphery of a toroid core. At the lower frequency end it is limited by the winding inductance and at the upper end by the coupling coefficient.

```
HYBTRAFO - Hybrid Transformer
* R1=R2=2R3=2R4
.AC DEC 60 1KHZ 1MEGHZ
.PRINT AC VDB(1) VDB(2)
.PRINT AC VDB(3) VDB(4)
V1 5 0 AC 2
R1 1 5 100
R2 2 0 100
R3 3 0 50
R4 4 0 50
L1 1 3 10MH
L2 3 2 10MH
L3 4 0 10MH
K12 L1 L2 .9998
K13 L1 L3 .9998
K23 L2 L3 .9998
.END
```





IsSpice AC Analysis of HYBTRAFO

6.1.3 Ratrace Coupler

The Ratrace coupler is an alternative to the hybrid coupler shown in 6.1.1. Here, the transmission lines are arranged circularly. Compared to the rectangular structure, this is an advantage as the electric field pattern remains undistorted. Reference lines, which determine the electric length of the transmission lines, take their course right in the center of the conductors and, therefore, allow an almost exact prediction of the circuit's behavior. With rectangular bends, more or less distinct discontinuities appear depending on the slimness and bending angle of the conductors. These causes a divergence from the theoretical model which has to be balanced by correction formulas [ref. 10] if precise simulation results are desired.




```
RATRACE - Lambda/4-Ratrace
V0 5 0 AC 2
R1 5 1 50
R2 2 0 50
R3 3 0 50
R4 4 0 50
T1 1 0 2 0 Z0=70.7 F=4.3GHZ NL=.25
T2 2 0 3 0 Z0=70.7 F=4.3GHZ NL=.25
T3 3 0 4 0 Z0=70.7 F=4.3GHZ NL=.25
T4 4 0 1 0 Z0=70.7 F=4.3GHZ NL=.75
.AC LIN 100 3.3GHZ 5.3GHZ
.PRINT AC VDB(1) VDB(2) VDB(3) VDB(4)
.END
```



IsSpice AC Analysis of RATRACE

6.1.4 Branchline Coupler

As the simulation in section 6.1.1 shows, the decoupling of node (4) from (1) is very narrow-banded. However, the bandwidth may be increased, to a certain degree, by using three lateral branches in place of two. This could be necessary in a balanced mixer application, for example, where an antenna is connected to port (1) and the local oscillator to port (4), with both sources being isolated from each other. Further increasing the number of branches isn't practicable because the impedance of the outside branches rises too high.




```
ARMKOPPL - Lambda/4 Branchline Coupler
V0 7 0 AC 2
R1 7 1 50
R2 2 0 50
R3 3 0 50
R4 4 0 50
T1 1 0 5 0 Z0=50 F=4.3GHZ NL=.25
T2 5 0 2 0 Z0=50 F=4.3GHZ NL=.25
T3 4 0 6 0 Z0=50 F=4.3GHZ NL=.25
T4 6 0 3 0 Z0=50 F=4.3GHZ NL=.25
T5 1 0 4 0 Z0=119 F=4.3GHZ NL=.25
T6 5 0 6 0 Z0=70 F=4.3GHZ NL=.25
T7 2 0 3 0 Z0=119 F=4.3GHZ NL=.25
.AC LIN 100 3.3GHZ 5.3GHZ
.PRINT AC VDB(1) VDB(2) VDB(3) VDB(4)
.END
```



IsSpice AC Analysis of ARMKOPP1